

Анализ предельных значений


В бизнесе очень часто приходится выявлять и анализировать предельные значения:

- Товары, которые лучше всего продаются
- Лучших клиентов
- Искать наименьшую цену на товар/услугу на рынке среди поставщиков и т.д.

Это можно делать разными способами, при наличии условий и ограничений и без.

В этой главе мы разберем:

- Как быстро находить наименьшие и наибольшие значения и выстраивать рейтинги с помощью функций **НАИБОЛЬШИЙ** и **НАИМЕНЬШИЙ**.
- Как искать максимальное и минимальное **по одному или нескольким условиям**.
- Как выявлять **самые часто встречающиеся** значения (для чисел и текста).
- Как выполнить **частотный анализ данных** одной формулой массива с функцией **ЧАСТОТА**.



Выявление топовых значений

Скорее всего вам известны простые функции **МИН** (MIN) и **МАКС** (MAX), позволяющие быстро найти минимальное или максимальное значение в таблице. Но что если нам требуется найти, например, не самое большое, а 2-е или 5-е значение в ТОПе? Здесь помогут функции **НАИБОЛЬШИЙ** (LARGE) и **НАИМЕНЬШИЙ** (SMALL). Синтаксис этих функций похож:

- **НАИБОЛЬШИЙ**(диапазон; позиция)
- **НАИМЕНЬШИЙ**(диапазон; позиция)

где

- **Диапазон** – диапазон ячеек с числами, которые мы проверяем.
- **Позиция** – целое число, представляющее собой позицию (ранг, номер в рейтинге) извлекаемого элемента.

Например, если у нас есть таблица с прибылями по товарам, то с помощью функции **НАИБОЛЬШИЙ** можно легко определить максимальное (первое в рейтинге) значение из диапазона прибыли (B2:B22), которое фактически будет тождественно формуле МАКС(B2:B22):

	A	B	C	D	E	F	G	H
1	Наименование	Прибыль						
2	Сыр	1 982			Лучшая прибыль		8270	
3	Сгущенка	530						
4	Макароны	5 785						
5	Масло	1 500						
6	Кукуруза	1 826						
7	Огурцы	796						
8	Соль	8 270						
9	Помидоры	5 235						
10	Черешня	4 860						
11	Фасоль	5 655						
12	Квас	658						

Аналогичным образом, функция

=НАИБОЛЬШИЙ(B2:B22;2)

... выдаст следующее за ним максимальное (2-е в рейтинге) и т.д.

Что интересно, аргумент **Позиция** может быть не просто числом. Так, например, для получения суммы первых трех максимальных значений в диапазоне можно использовать формулу массива с прописанным внутри массивом констант для первых трех позиций (сочетание **Ctrl+Shift+Enter** в конце можно не нажимать):

G4		={СУММ(НАИБОЛЬШИЙ(B2:B22;{1;2;3}))}						
	A	B	C	D	E	F	G	H
1	Наименование	Прибыль						
2	Сыр	1 982				Лучшая прибыль	8 270	
3	Сгущенка	530						
4	Макароны	5 785				Сумма прибыли трех лучших товаров	22 285	
5	Масло	1 500						
6	Кукуруза	1 826						
7	Огурцы	796						
8	Соль	8 270						
9	Помидоры	5 235						

Если же нам нужно извлечь не числовые значения, а соответствующие им названия товаров (имена клиентов), то придется использовать более сложную конструкцию, похожую на разобранный ранее пример с «левым ВПР» (я использовал перенос строк с помощью **Alt+Enter** для наглядности):

```
{=ИНДЕКС($A$5:$A$25;
    ПОИСКПОЗ(НАИБОЛЬШИЙ($B$5:$B$25;G10);
        $B$5:$B$25;
        0))}
```

H10		={ИНДЕКС(\$A\$5:\$A\$25;ПОИСКПОЗ(НАИБОЛЬШИЙ(\$B\$5:\$B\$25;G10);\$B\$5:\$B\$25;0))}												
	A	B	C	D	E	F	G	H	I	J	K	L		
1	Топ-5 лучших товаров													
2	Поиск N максимальных или минимальных значений в списке и их суммы													
3														
4	Наименование	Прибыль												
5	Сыр	1 982				Лучшая прибыль	9 539							
6	Сгущенка	530												
7	Макароны	5 785				Сумма прибыли трех лучших товаров	26 000							
8	Масло	1 500												
9	Кукуруза	1 826												
10	Огурцы	796				Топ лучших товаров	1	Мука						
11	Соль	8 270					2	Соль						
12	Помидоры	5 235					3	Чай						
13	Черешня	4 860					4	Гречка						
14	Фасоль	5 655					5	Макароны						
15	Квас	658												
16	Кисель	2 200												

В английской версии это будет:

```
{=INDEX($A$5:$A$25;MATCH(LARGE($B$5:$B$25;G10);$B$5:$B$25;0))}
```

Разберем эту формулу по частям:

- **НАИБОЛЬШИЙ(\$B\$5:\$B\$25;G10)** – ищет наибольшее значение с заданным рейтингом (1) в столбце *Прибыль*.
- **ПОИСКПОЗ(НАИБОЛЬШИЙ(\$B\$5:\$B\$25;G10);\$B\$5:\$B\$25;0)** – определяет порядковый номер ячейки в столбце *Прибыль*, где лежит наибольшее значение (это будет число 17, т.к. наибольшее значение 9539 лежит в 17-й по счету ячейке в столбце *Прибыль*).
- **ИНДЕКС(\$A\$5:\$A\$25;ПОИСКПОЗ(НАИБОЛЬШИЙ(\$B\$5:\$B\$25;G10);\$B\$5:\$B\$25;0))** – выдает содержимое 17-й ячейки из столбца *Наименования*, т.е. *Мука*.

Поиск минимального/максимального по условию

В Microsoft Excel давно есть в стандартном наборе функции **СЧЁТЕСЛИ** (COUNTIF), **СУММЕСЛИ** (SUMIF) и **СРЗНАЧЕСЛИ** (AVERAGEIF) и их аналоги **СЧЁТЕСЛИМН** (COUNTIFS), **СУММЕСЛИМН** (SUMIFS) и **СРЗНАЧЕСЛИМН** (AVERAGEIFS), позволяющие искать количество, сумму и среднее в таблице по одному или нескольким условиям. Но только в июньском обновлении последней версии Microsoft Excel 2016 появились функции, которые умеют искать по условию максимум и минимум - они называются **МАКСЕСЛИМН** (MAXIFS) и **МИНЕСЛИМН** (MINIFS). Но что же делать, если у вас не последняя версия Excel?

Чтобы обойти это ограничение и решить проблему, можно использовать небольшую формулу массива.

Предположим, нам нужно найти минимальную цену для каждого товара в базе данных по поставщикам:

The screenshot shows an Excel spreadsheet with the following content:

- Formula bar: `{=МИН(ЕСЛИ(A7=F7:F34;H7:H34))}`
- Cell B7: 127
- Table 1 (Columns: Товар, Мин.цена, Поставщик):

Товар	Мин.цена	Поставщик
Чай	127	Марунга
Кофе	250	Траск
Сахар	34	Жар-птица
- Table 2 (Columns: Товар, Поставщик, Цена):

Товар	Поставщик	Цена
Сахар	Василь и К	83
Сахар	Траск	61
Сахар	Кобальт	52
Сахар	Альфа	97
Чай	Оберег	442
Сахар	Жар-птица	34
Кофе	Траск	250
Сахар	Марунга	46
Чай	Василь и К	201
Чай	Марунга	127
Сахар	Мереж	78
Чай	Мереж	475
Кофе	Жар-птица	695

В английской версии это будет соответственно

`{=MIN(IF(A7=F7:F34;H7:H34))}`

Разберем логику работы этой формулы.

Функция **ЕСЛИ** проверяет каждую ячейку массива из столбца *Товар* (F7:F34) на предмет равенства текущему товару. Если это так, то выдается соответствующее ему значение из столбца *Цена* (H7:H34). В противном случае – логическое значение ЛОЖЬ, которое игнорируется потом функцией **МИН**.

Для наглядности можно выделить мышью всю функцию ЕСЛИ(...) и нажать на клавиатуре **F9**, чтобы увидеть результирующий массив, из которого потом функция МИН и выбирает минимальное значение:

С7 : X ✓ fx {=ИНДЕКС(\$G\$7:\$G\$34;ПОИСКПОЗ(A7&B7;\$F\$7:\$F\$34&\$H\$7:\$H\$34;0))}

Выбор лучших поставщиков						
Выбор поставщиков с минимальной ценой						
Заказ				База данных по поставщикам		
Товар	Мин.цена	Поставщик	Товар	Поставщик	Цена	
Чай	127	Марунга	Сахар	Василь и К	83	
Кофе	250	Траск	Сахар	Траск	61	
Сахар	34	Жар-птица	Сахар	Кобальт	52	
			Сахар	Альфа	127	
			Чай	Оберег	442	

Она работает аналогично предыдущей, но ищет склеенную с помощью оператора конкатенации «&» пару Чай127 (A7&B7) в склеенных столбцах ТоварЦена (\$F\$7:\$F\$34&\$H\$7:\$H\$34).